
Suggested changes to PackageInfo-Exporters 
 
 
 
 

1. Need to include the Seaside namespace when naming 
classes 
 
1.1. A typical example is the export of a class that inherits from WAComponent. The superclass name that 
is supplied for the class definition has to be Seaside.WAComponent. 
 
Object class 
 
visualworksName 
 (self name copyFrom: 1 to: 2) = 'WA' ifTrue: [^'Seaside.', self name]. 
 ^ self == Object ifTrue: ['Core.Object'] ifFalse: [self name] 
 
1.2. Another situation occurs when exporting a package that is adding method(s) to some Seaside 
classe(s). The class name for those added methods must be prefixed with the Seaside namespace. 
 
Behavior 
 
visualworksName 
 ^ (self name copyFrom: 1 to: 2) = 'WA' 
  ifTrue: ['Seaside.', self name] 
  ifFalse: [self name] 
 



2. Need to export class comments 
 
When class comments are available in Squeak, they should be exported. 
 
VW5PackageExporter 
 
fileOutClassCommentForClass: aClass 
  
 aClass hasComment ifFalse: [^self]. 
 stream cr. 
 self tag: 'comment' do: [ 
  self tag: 'class-id' with: (self fullyQualifiedClassNameFor: aClass). 
  self tag: 'body' with: aClass comment asString. 
 ] 
 
fileOutClassDefinition: aClass 
 stream cr. 
  
 self tag: 'class' do: [ 
  self tag: 'name' with: aClass name. 
  self tag: 'environment' with: self environment. 
  self tag: 'super' with: (self fullyQualifiedClassNameFor: aClass superclass). 
  self tag: 'private' with: 'false'. 
  self tag: 'indexed-type' with: (self mapType: aClass typeOfClass). 
  self tag: 'inst-vars' with: aClass instanceVariablesString. 
  self tag: 'class-inst-vars' with: aClass class instanceVariablesString. 
  self tag: 'imports' with: ''. 
  self tag: 'category' with: aClass category. 
  self tag: 'attributes' do: [ 
   self tag: 'package' with: (self packageNameForClass: aClass). 
  ] 
 ]. 
  
 aClass classPool keys do: [:ea | self fileOutClassVariable: ea class: aClass]. 
  
 self fileOutClassCommentForClass: aClass 
 



3. Need to import the Seaside namespace 
 
When an exported package subclasses some Seaside classes, the Seaside namespace should be 
imported by the exported package namespace. 
 
 
VW5PackageExporter 
 
fileOutNameSpace 
 
 self tag: 'name-space' do: [ 
  self tag: 'name' with: self environment. 
  self tag: 'environment' with: 'Smalltalk'. 
  self tag: 'private' with: 'false'. 
  self tag: 'imports' do: [ 
   stream nextPutAll: self nameSpaceImports; cr. 
  ]. 
  self tag: 'category' with: self environment. 
  self tag: 'attributes' do: [ 
   self tag: 'package' with: self environment. 
  ]. 
 ] 
 
nameSpaceImports 
 
 | imports result | 
 imports := Set new. 
 package classes do: 
  [:ea | 
  (ea name copyFrom: 1 to: 2) = 'WA' 
   ifTrue: [imports add: #Seaside]. 
  (ea superclass name copyFrom: 1 to: 2) = 'WA' 
   ifTrue: [imports add: #Seaside]]. 
 result := String new writeStream. 
 result nextPutAll: 'Smalltalk.*'. 
 imports do: [:ea | result space; nextPutAll: ea; nextPutAll: '.*']. 
 ^ result contents 
 


